Programmation Basic avec StarOffice

Star Office dispose d’un IDE permettant de définir des procédures et des fonctions en langage BASIC. 

Par ailleurs, il est possible avec un document Star Office d’utiliser des objets graphiques pour l’interaction avec l’utilisateur.

Documents de référence 

· OpenOffice.org Developer Guide and API Reference (1065 pages mais vraiment utile, chapitre 11 sur le langage Basic et les boîtes de dialogue notamment) : http://api.openoffice.org/docs/DevelopersGuide/DevelopersGuide.pdf
· Manuel de programmation Basic StarOffice (230 pages) : http://docs-pdf.sun.com/817-3917/817-3917.pdf

Accès à l’IDE de StarOffice

Menu Outils/Macros/Macro

Cet IDE ne propose pas d’assistance lors de la définition et de l’accès aux objets.

Le code des procédures et fonctions est stocké dans des modules.

Pour tester les procédures d’un module, il faut se placer sur le module et utiliser le bouton   

 dans la barre d’outils 

Notion de formulaire

Le concept de formulaire Star Office est similaire de celui en HTML. Il s’agit d’un ensemble de contrôles graphiques qui sont insérés dans un document, et qui permettent une interaction avec l’utilisateur.

Dans ce qui suit, le document est une feuille Calc.

Un contrôle graphique est appelé élément de contrôle. Pour pouvoir placer des contrôles, il faut faire apparaître la barre de fonctions de formulaire :

Formulaire de base de données

Le contrôle de table placé sur document Star Office permet d’obtenir une grille comme ci-dessous. Un assistant (Autopilote) permet de paramétrer la grille.

Création d’un contrôle graphique non lié

Ce contrôle fait partie d’un formulaire que nous avons nommé formDevis.

Boîtes de dialogue

Les boîtes de dialogue permettent une interactivité avec l’utilisateur indépendamment du document. Les boîtes de dialogue, comme les formulaires, comprennent des contrôles. 

Pour créer une boîte de dialogue sous l’IDE, on clique sur la barre d’onglets puis Insérer/Boîte de dialogue.

Pour faire apparaître la palette des contrôles, on clique sur le bouton suivent de la barre d’outils :

Une boîte de dialogue est associée à deux objets : la vue qui est la partie visible du composant et le modèle qui permet d’accéder au contenu du composant.

L’ouverture de boîtes de dialogue est prise en charge par une procédure :

Sub Dialog1Show

 public oDialog1 as object ‘La boîte de dialogue

 public oDialog1Model as object ‘Le modèle 

 DialogLibraries.LoadLibrary( "Standard" )

 oDialog1 = CreateUnoDialog( DialogLibraries.Standard.choixClient2 )

 oDialog1Model = oDialog1.Model 

 ‘ Paramétrage des propriétés du modèle

 oDialog1Model.Title="Choix du client"

 oDialog1Model.Height=50

 oDialog1Model.Width=100

 oDialog1Model.PositionX=250

 oDialog1Model.PositionY=100

End Sub

Le contrôles sont les mêmes que pour les formulaires, l’accès à un contrôle placé sur une boîte de dialogue s’effectue de la façon suivante :

 Dim oListBox  as object ‘ ici c’est une zone de liste

 oListBox = oDialog1.GetControl("lbClient")

On peut ensuite manipuler l’objet contrôle, par exemple récupérer le nombre d’éléments de la zone de liste et ajouter un élément :

     Dim iCount as integer

  iCount = oListbox.ItemCount

  oListbox.additem(ident, iCount) 

Pour que le premier élément de la liste soit sélectionné par défaut :

  oListBox.selectItemPos(0,1)  ‘0 : position du premier, 1 valeur booléenne "vrai" pour sélectionner

Pour les propriétés et les méthodes d’une ListBox, voir Voir aussi http://api.openoffice.org/docs/common/ref/com/sun/star/awt/XListBox.html.

Pour les propriétés des contrôles, voir le manuel Star Office 7 de programmation Basic chapitre 11 à partir de la page 189.

Accès à une base de données 

Le langage Basic de Star Office permet d’instancier des objets techniques pour l’accès aux bases de données. Pour accéder à une base de données, nous retiendrons qu’il faut

· instancier un objet conteneur de toutes les sources de données qui seront utilisées lors de la session de Star Office (que l’on nommera databaseContext) ;

· instancier un objet pour les caractéristiques d’une source de données (que l’on nommera dataSource) ;

· instancier un objet pour la connexion à la base de données (connection) ;

· instancier un objet pour envoyer une requête SQL à la base de données (statement)

· instancier un objet pour le jeu d’enregistrements récupéré d’une table ou d’une requête (resultSet) ;

Les procédures suivantes présentes les opérations typiques sur une base de données 

Sub Main

' Procédure de connexion à la source de données

  Dim databaseContext as object 'objet racine pour toutes les opérations sur les bases de données

  Dim dataSource As Object 'objet associé à une source de données donc à une base de données

  Dim connection as object 'canal de connexion à la source de données

  databaseContext=createUnoService("com.sun.star.sdb.DatabaseContext") 'création du conteneur racine

  dataSource=databaseContext.getByName("devis") 'création de l'objet source de données pour l'application

  connection=dataSource.GetConnection("root","") 'activation de la connexion à la base de données

End Sub

Sub ParcoursClient

' Procédure de visualisation de données 

  Dim statement as object 'objet pour représenter une requête

  Dim resultset as object 'objet stockant le résultat d'une requête ou le contenu d'une table

  Dim ident as string  ' pour la construction de la chaîne à afficher

  statement=connection.createStatement()

  resultset=statement.executeQuery("SELECT numClient,nomClient, prenomClient FROM Client")

  If Not IsNull(resultset) Then  'Si la requête a renvoyé des données

    While resultset.next 'Parcours du flot d'enregistrements

      ident = resultSet.getString(3)+ " " + resultSet.getString(2)  'Concaténation de chaînes

      'Récupération du contenu de la deuxième et de la troisième colonne de l'enregistrement courant

      'Ici le type de chaque donnée à récupérer est une chaîne donc on utilise la méthode getString

      MsgBox ident  ' Affichage des données concaténées d'une ligne du resulSet

    Wend

  End If  

End Sub

Sub RechClientNum

' Procédure de recherche d'une ligne dans une table à partir d'un champ numérique

  Dim leNum as integer

  Dim txtReq as string

  Dim statement1 as object 

  Dim resultset1 as object

  leNum=1 ' Valeur de champ pour la sélection

  statement1=connection.createStatement()

  txtReq="SELECT adrRueClient, adrVilleClient, CPClient FROM Client WHERE numClient = "+leNum

  resultSet1=statement1.executeQuery(txtReq)

  If Not IsNull(resultset1) Then

    resultset1.next

    MsgBox resultSet1.getString(3) 'Affichage d'un champ de la ligne sélectionnée

  End If

End Sub 

Sub RechClientNom

' Procédure de recherche d'une ligne dans une table à partir d'un champ chaîne

  Dim leNom as string

  Dim txtReq as string

  Dim statement1 as object 

  Dim resultset1 as object

  leNom="Séré" 'Valeur pour la sélection d'une ligne

  statement1=connection.createStatement()

  txtReq="SELECT adrRueClient, adrVilleClient, CPClient FROM Client WHERE nomClient = '"+leNom+"'"

  resultSet1=statement1.executeQuery(txtReq)

  If Not IsNull(resultset1) Then

    resultset1.next

    MsgBox resultSet1.getString(3) 'Affichage d'un champ de la ligne sélectionnée

  End If

End Sub 

Sub MAJClient

' Procédure de mise à jour d’un champ dans une ligne d'une table

  Dim leNom as string

  Dim leCP as string

  Dim result as long

  Dim txtReq as string

  Dim statement1 as object 

  Dim resultSet1 as object

  leNom="Séré" ' Valeur pour la sélection de la ligne

  leCP="21000" ' valeur de champ à mettre à jour

  statement1=connection.createStatement()

  txtReq="UPDATE Client SET CPClient='"+leCP+"' WHERE nomClient = '"+leNom+"'"

  result=statement1.executeUpdate(txtReq) 'la méthode executeUpdate renvoie un entier long

  If result <> 1 then

    msgBox "problème de MAJ"

  End If

End Sub   

Utilisation d’un objet resultSet 

Un objet resultSet peut être instancié pour récupérer une liste (voir plus haut) mais il est aussi possible de faire des recherches dans cet objet. Car une fois chargé en mémoire et visible (déclaré public par exemple), il permet à plusieurs procédures de l’utiliser sans  réinterroger la base de données.

Si on veut pouvoir parcourir le resultSet dans plusieurs sens, il faut paramétrer une propriété de l’objet statement qui va permettre l’instanciation de l’objet resultSet :

  statement=connection.createStatement()

  statement.ResultsetType=1004 'SCROLL_INSENSITIVE, autorise tout type de navigation dans le resultSet

La propriété ResultsetType si elle n’est pas paramétrée est par défaut égale à 1003 (FORWARD_ONLY). Une autre valeur possible est SCROLL_SENSITIVE pour permettre que les changements des données d’origine modifient le ResultSet. 

Une autre propriété permet de spécifier si les données du resultSet peuvent être modifiées : ResultSetConcurrency. Par défaut cette propriété est READ_ONLY (l’autre valeur est UPDATABLE).

Ensuite l’objet resultSet possède plusieurs méthodes pour se déplacer. La méthode absolute(valeur) permet de se positionner sur une ligne du resultSet dont on indique la position. Voir page 828 du guide du développeur pour les autres méthodes.

L’exemple suivant montre l’utilisation d’un objet resultSet dans une boîte de dialogue :

Ici deux procédures évènementielles sont définies, elles partagent les contrôles et la connexion à la base de données, donc les objets déclarés pour ces fonctions sont déclarés publics.

public oDialog1 as object

public oDialog1Model as object

public oListBox as object 'contrôle de la boîte

public databaseContext as object 'objet racine pour toutes les opérations sur les bases de données

public dataSource As Object 'objet associé à une source de données donc à une base de données

public connection as object 'canal de connexion à la source de données

public statement as object 'objet pour représenter une requête

public resultset as object 'objet stockant le résultat d'une requête ou le contenu d'une table

'procédure d'affichage de la boîte de dialogue

Sub Dialog1Show

  DialogLibraries.LoadLibrary( "Standard" )

  oDialog1 = CreateUnoDialog( DialogLibraries.Standard.choixClient )

  oDialog1Model = oDialog1.Model 'accès au modèle d'une boîte de dialogue, pour accéder au contenu

  oDialog1Model.Title="Choix du client"

  'oDialog1.Model.Title= "Choix du client" ; autre écriture

  oListBox = oDialog1.GetControl("lbClient") 'accès à la zone de liste définie sur la boîte de dialogue

' Récupération des noms et prénoms des clients

' Procédure de connexion à la source de données

  databaseContext=createUnoService("com.sun.star.sdb.DatabaseContext") 'création du conteneur racine

  dataSource=databaseContext.getByName("devis") 'création de l'objet source de données pour l'application

  connection=dataSource.GetConnection("root","") 'activation de la connexion à la base de données

' Procédure de visualisation des prénoms et noms de tous les clients

  Dim ident as string  ' pour la construction de la chaîne à afficher

  statement=connection.createStatement()

  statement.ResultsetType=1004 'SCROLL_INSENSITIVE, autorise tout type de navigation dans le resultSet

  resultset=statement.executeQuery("SELECT numClient, nomClient, prenomClient, adrRueClient, adrVilleClient, CPClient FROM Client")

  dim iCount as integer

  iCount = oListbox.ItemCount

  If Not IsNull(resultset) Then  'Si la requête a renvoyé des données

    While resultset.next 'Parcours du flot d'enregistrements

      ident = resultSet.getString(3)+ " " + resultSet.getString(2)  'Concaténation de chaînes

      'Récupération du contenu de la deuxième et de la troisième colonne de l'enregistrement courant

      'Ici le type de chaque donnée à récupérer est une chaîne donc on utilise la méthode getString

      iCount = oListbox.ItemCount

      oListbox.additem(ident, iCount)  ' Affichage des données concaténées dans la zone de liste

     Wend

  End If   

oDialog1.Execute()

End Sub

' procédure suite à sélection d'un client

Sub Client_Click

  Dim oAdrRue, oAdrVil, oNumCli as object

  oAdrRue= oDialog1.GetControl("adrRueClient")

  oAdrVil= oDialog1.GetControl("adrVilClient")

  oNumCli= oDialog1.GetControl("lNum")

  resultSet.absolute(oListBox.selectedItemPos+1) 'on va dans le resultSet récupérer la ligne correspondant à la sélection 

  ' on affiche dans la zone de texte les informations sélectionnées

  oNumCli.Text=resultSet.getString(1)

  oAdrRue.Text=resultSet.getString(4)

  oAdrVil.Text=resultSet.getString(6)+" "+resultSet.getString(5)

End sub

Accès aux éléments d’un classeur

Le langage Basic permet accéder aux éléments d’un classeur (feuilles et cellules) en manipulant des objets techniques.

Accès à une feuille de calcul

On peut accéder aux feuilles d’un classeur à travers la liste des feuilles sheets. Une feuille peut être désignée par sa position dans la liste ou par son nom (dans l’exemple suivant la feuille se nomme « devisAppli »). 

Sub accesFeuille

‘ Procédure d’accès à une feuille

  Dim doc as object 'le document courant

  Dim feuille as object 'la feuille courante

  doc=ThisComponent 'le document est celui qui demande l'exécution de ce code

 'feuille=doc.Sheets(1) 'Sheets est la liste des feuilles indicée à partir de 0, la deuxième feuille nous intéresse

  feuille=doc.Sheets.getByName("devisAppli") 'on adresse la feuille par son nom

End Sub

Accès à une cellule

On peut accéder aux cellules d’une feuille en utilisant une méthode de la feuille soit en désignant la position de la cellule dans la feuille soit en donnant son nom. 

Selon le contenu de la cellule, on utilisera une des trois propriétés suivantes d’un objet représentant une cellule :

· Value pour un nombre,

· String pour une chaîne,

· Formula pour une formule.

Sub AccesCell

  Dim doc as object 'le document courant

  Dim feuille as object 'la feuille courante

  Dim cell, cellDate as object 'une cellule

  doc=ThisComponent 'le document est celui qui demande l'exécution de ce code

 'feuille=doc.Sheets(1) 'Sheets est la liste des feuilles indicée à partie de 0, la deuxième feuille nous intéresse

  feuille=doc.Sheets.getByName("devisAppli") 'on adresse la feuille par son nom

 ' cell=feuille.getCellByPosition(0,0) 'on adresse la cellule A1 par sa position

  cell=feuille.getCellRangeByname("A1") 'on adresse la cellule A1 par son nom

  msgBox cell.string 'on affiche le contenu de la cellule dont on sait que c'est une chaîne

  cellDate=feuille.getCellRangeByname("E1")

  cellDate.string=Date  ' on met la date du jour dans la cellule

End Sub

Remarque : Il est possible sous Calc de donner un nom mnémonique à une cellule : Menu/ Insertion/Nom/Définir.

Accès à un contrôle d’un formulaire placé sur une feuille

Un formulaire placé sur une feuille de calcul est accessible par un objet Dessin associé à la feuille de calcul.

Sub accesControle

' Accède à une listBox d'un formulaire placé sur une feuille

  Dim doc as object 'le document courant

  Dim feuille as object 'la feuille courante

  Dim dessinFeuille as object 'le dessin de le feuille

  Dim form as object 'le formulaire

  Dim listBox as object ' la listBox qui fait partie du formulaire

  doc=ThisComponent 'le document est celui qui demande l'exécution de ce code

  feuille=doc.Sheets.getByName("devisAppli") 'on adresse la feuille par son nom

  dessinFeuille=feuille.Drawpage 'on accède au dessin de la feuille

  form=dessinFeuille.Forms.GetByIndex(0) 'on adresse le premier formulaire de la feuille par son dessin

  listBox=form.getByName("lbClient") 'on accède au contrôle par son nom

  msgBox listBox.name

End Sub

Association de procédures à des événements sur des contrôles 

Les contrôles sont capables de réagir à des événements prédéfinis. La réaction à un événement est programmée à l’aide d’une procédure associée à l’événement.

Les procédures sont stockées dans des modules.

Ouverture d’une boîte de dialogue lors de l’accès à une feuille de calcul

Menu Outils/Adaptation : on choisit l’événement d’une part, le module et la procédure qui devra s’exécuter d’autre part.

Attention si deux modules ont des procédures de même entête, c’est la première procédure définie qui sera lancée même si l’association entre le contrôle et la procédure a été spécifiée pour le bon module (.

ATTENTION : l’association d’une procédure à un événement ne fonctionne pas très bien (voir http://www.oooforum.org/forum/viewtopic.php?t=5625)

Il est nécessaire de :

· nommer le fichier de l’application sans extension,

· effectuer l’association de la procédure à l’événement,

· renommer le fichier avec l’extension.

Association d’une procédure à un événement sur un contrôle

En se plaçant sur un contrôle, le click droit fait apparaître un menu contextuel, choisir Propriétés. L’onglet Evènements liste tous les événements possibles. On sélectionne l’événement et on choisit le module et la procédure à associer. 

La procédure Client2_Click permet, à partir d’un client sélectionné dans la zone de liste, de renseigner des cellules d’une feuille de calcul.

Sub Client2_Click

  Dim oAdrRue, oAdrVil, oNumCli as object

  resultSet.absolute(oListBox.selectedItemPos+1) 'on va dans le resultSet récupérer la ligne correspondant à la sélection 

  ' on affiche dans les cellules les informations sélectionnées

  Dim cellIdent, cellAdrRue, cellAdrVil as object 'cellules à renseigner

  cellIdent=feuille.getCellRangeByname("B6") 'on adresse la cellule par son nom

  cellAdrRue=feuille.getCellRangeByname("adrRueClient") 'on adresse la cellule par son nom

  cellAdrVil=feuille.getCellRangeByname("adrVilClient") 'on adresse la cellule par son nom

  cellIdent.String=oListBox.selectedItem ‘'l'élément sélectionné dans la zone de liste sert pour renseigner cette cellule

  cellAdrRue.String=resultSet.getString(4)

  cellAdrVil.String=resultSet.getString(6)+" "+resultSet.getString(5)

End sub

 Pour mieux comprendre le contexte de cette procédure, voir plus haut le paragraphe « Utilisation d’un objet resultSet » . 







A l’ouverture de la boîte de dialogue, la zone de texte lbClient est alimentée avec la liste des noms et prénoms des clients.





Lorsque l’utilisateur sélectionne un nom, son adresse est affichée dans les deux zones de texte.





Contrôle de classe ListBox


nommé


lbClient





La désactivation du mode conception permet la visualisation des données





Permet de passer du mode conception au mode exécution pour tester les contrôles d’un formulaire.





Cliquer/ Déplacer sur cet icône pour faire apparaître la barre de fonctions de formulaire





Contrôle de table





Ce bouton permet d’accéder à l’IDE et de visualiser le code des modules et les boîtes de dialogue définis dans l’application.

